

TRANSFORMASI DATA SKALA ORDINAL KE INTERVAL DENGAN MENGGUNAKAN MAKRO MINITAB

The Transformation of Ordinal Data to Interval Data Using Makro Minitab

Budi Waryanto¹ and Yuan Astika Millafati²

1. Pusat Data dan Informasi Pertanian

2. Mahasiswi Program Studi Statistika, IPB.

ABSTRACT

Ordinal Data is frequently used in path analysis. In that case, the ordinal data is often transformed to an interval data. The transformation process needs a conversion program, which is easy and fast. Makro minitab can be used in that transformation, however this program has a weakness. This article will discuss the improvement, which has been made in the transformation of ordinal data to an interval data at least with data scale three (1-3, 1-5, 1-7, etc). In order to perform that task, a transformation program, namely "Konverum" is designed by using the makro Minitab. This program is able to transform ordinal data in Likert scale to an interval data, at least with scale 3 category. In addition, this program is also capable in transforming data which is not completely filled out.

PENDAHULUAN

Analisis data menggunakan teknik statistik selama ini lebih dikenal untuk data kuantitatif yang sering dijumpai pada penelitian ilmu-ilmu eksakta seperti bidang ekonomi, pertanian, biologi dan bidang ilmu kuantitatif lainnya. Pada penelitian di bidang ilmu sosial seringkali digunakan data kualitatif sebagai refleksi dari konsep yang sifatnya abstrak, atau tidak bisa diukur secara langsung. Oleh karena itu analisis yang sering digunakan terbatas pada analisis diskriptif atau analisis non parametrik, jarang sekali menggunakan analisis parametrik seperti analisis regresi atau sejenisnya. Akan tetapi pada saat ini penelitian-penelitian di bidang ilmu sosial sudah berkembang dan banyak dijumpai pemakaian analisis kuantitatif, walaupun peubahnya bersifat kualitatif. Salah satu analisis yang berkembang saat ini adalah analisis jalur (*path analysis*).

Pada pendekatan analisis jalur seringkali digunakan tipe data ordinal. Tipe data tersebut merefleksikan peubah-peubah yang sebelumnya berasal dari suatu konsep yang telah diubah bentuknya sehingga dapat diukur (Nazir, 1988). Analisis jalur membutuhkan perhitungan matematis di dalamnya. Oleh karena itu skala pengukuran data yang dibutuhkan minimal berskala interval. Jika data yang akan dianalisis berskala ordinal, maka perlu ditransformasi terlebih dulu menjadi skala interval agar dapat digunakan untuk analisis lebih lanjut.

Hasil kajian tentang proses transformasi data dari tipe ordinal ke dalam bentuk tipe interval sebagai prasarat sebelum dilakukan analisis jalur belum banyak dipublikasikan. Pengkajian pada permasalahan tersebut pernah dilakukan oleh Muchlis (2001) dengan menggunakan bantuan program makro Minitab. Akan tetapi program untuk melakukan proses transformasi tersebut masih belum sempurna, karena hanya bisa melakukan proses transformasi untuk *skala likert* dengan jumlah pilihan sebanyak lima (biasanya merupakan jawaban responden pada suatu penelitian sosial yang diindikasikan dengan *skala likert* sebagai berikut: 1=Sangat Tidak Setuju, 2=Tidak Setuju, 3=Netral, 4=Setuju dan 5=Sangat Setuju) dan semuanya harus terisi. Jawaban pada skala 1-5 tersebut pada proses analisis data dimasukkan pada jenis data ordinal. Berdasarkan hal tersebut, tulisan ini akan memfokuskan pada perbaikan proses transformasi yang sebelumnya telah dilakukan. Tujuannya adalah memperbaiki program transformasi data ordinal ke data interval menggunakan makro Minitab yang berlaku umum. Adapun manfaat yang diperoleh dari tulisan ini diharapkan dapat membantu para analis/peneliti sosial dalam melakukan pra analisis yakni transformasi data ordinal ke interval sebelum dilakukan analisis inferensia lebih lanjut.

Salah satu cara yang paling sering digunakan dalam menentukan skor adalah dengan menggunakan skala Likert. Skala Likert adalah ukuran gabungan yang didasarkan pada struktur intensitas pertanyaan-pertanyaan. Dengan demikian, skala Likert sebenarnya bukan skala, melainkan suatu cara yang lebih sistematis untuk memberi skor pada indeks (Singarimbun dan Effendi, 1995).

METODOLOGI

Data yang digunakan berasal dari hasil penelitian tentang kajian Implementasi Kebijakan Penyuluhan Pertanian di Kabupaten Subang. Dengan jumlah indikator sebanyak 64 buah, empat sub variabel bebas (x_i) yang mencerminkan dimensi Implementasi Kebijakan (X), terdiri dari :

X1= Komunikasi
X2= Sumberdaya
X3= Disposisi
X4= Struktur birokrasi

serta tiga sub variabel tak bebas (y_i) yang mencerminkan dimensi Kemandirian Kelompok Tani (Y), terdiri dari :

Y1= Kemandirian Sosial
Y2= Kemandirian Ekonomi
Y3= Kemandirian Politik.

Metode transformasi yang digunakan yakni *method of successive interval*, Hays (1976), dengan bantuan makro minitab. Metode tersebut digunakan untuk melakukan transformasi data ordinal menjadi data interval. Pada umumnya jawaban responden yang diukur dengan menggunakan skala likert (*Lykert scale*) diadakan *scoring* yakni pemberian nilai numerikal 1, 2, 3, 4 dan 5, setiap skor yang diperoleh akan memiliki tingkat pengukuran ordinal. Nilai numerikal tersebut dianggap sebagai objek dan selanjutnya melalui proses transformasi ditempatkan ke dalam interval.

Langkah-langkahnya sebagai berikut:

1. Untuk setiap pertanyaan, hitung frekuensi jawaban setiap kategori (pilihan jawaban).
2. Berdasarkan frekuensi setiap kategori dihitung proporsinya.
3. Dari proporsi yang diperoleh, hitung proporsi kumulatif untuk setiap kategori.
4. Tentukan pula nilai batas Z untuk setiap kategori.

5. Hitung *scale value* (interval rata-rata) untuk setiap kategori melalui persamaan berikut:

$$Scale = \frac{\text{kepadatan batas bawah} - \text{kepadatan batas atas}}{\text{daerah di bawah batas atas-daerah di bawah batas bawah}}$$

6. Hitung *score* (nilai hasil transformasi) untuk setiap kategori melalui persamaan:

$$score = scaleValue + |scaleValue_{min}| + 1$$

(Hays, 1976).

HASIL DAN PEMBAHASAN

Dari sekian banyak perangkat lunak pengolah data, pada umumnya belum ada yang khusus dapat digunakan untuk membantu melakukan proses *scalling*. Perhitungan secara manual mudah dilakukan, tetapi jika dihadapkan pada jumlah item pertanyaan yang cukup besar membutuhkan waktu perhitungan yang cukup lama dan menjadi tidak efisien.

Untuk memudahkan dan mempercepat proses transformasi data digunakan alat bantu komputer. Kendala yang dihadapi adalah tidak tersedianya pilihan khusus untuk melakukan transformasi data dalam perangkat lunak statistik saat ini (SPSS, SAS, Minitab, dan lainnya). Untuk keperluan tersebut, biasanya peneliti membuat program khusus melalui bahasa pemrograman tertentu.

Program yang dibuat khusus untuk *scalling* yang pertama kali dipublikasikan adalah STAT25 (Dachlan, 2001), melalui bahasa pemrograman Pascal. Namun program ini masih menghadapi banyak kendala. Input data berupa nilai frekuensi setiap item pertanyaan, sedangkan outputnya adalah *scale value* dari setiap item. Untuk memperoleh hasil akhir masih diperlukan dua langkah pekerjaan lagi, yaitu menghitung *score* (nilai hasil transformasi) untuk setiap kategori pada setiap itemnya dan merubah keseluruhan data pengamatan ke dalam skala interval.

Selanjutnya Kudus (1996) dalam Dachlan (2001), membuat program makro Minitab yang disebut 'konversi' dalam rangka mengatasi kendala yang ada pada STAT25. Prosedur perhitungannya menggunakan input proporsi kumulatif yang selalu berada pada kolom C1 dan output yang dihasilkan adalah *score* hasil *scalling*. Program ini dibuat untuk mengkonversi data ordinal ke data interval dengan jumlah pilihan lima buah untuk masing-masing item.

Selanjutnya Dachlan (2001) mengembangkan program makro minitab 'SCL_Ver1'. Input data ini adalah data mentah (hasil jawaban

responden) dalam skala ordinal, sedangkan outputnya adalah data hasil konversi data keseluruhan dalam skala pengukuran interval. Program ini dirancang untuk mengkonversi data ordinal ke interval dengan jumlah pilihan lima buah untuk setiap item. Listing program terlihat pada Lampiran 1.

Program ini juga masih memiliki kelemahan, diantaranya harus selalu dijalankan pada program Minitab. Kelemahan lainnya, program ini dibuat sebatas untuk melakukan transformasi data dengan lima pilihan jawaban dan untuk setiap item pertanyaan frekuensi pilihan jawaban harus terisi seluruhnya.

Oleh karena dari ketiga program untuk konversi data tersebut masih juga ditemukan kelemahan-kelemahan, maka perlu dilakukan penyempurnaan lebih lanjut. Berikut hasil perbaikan program makro Minitab seperti telah disebutkan pada tujuan sebelumnya yang juga merupakan penjabaran dari *method of successive interval*. Program makro minitab ini dinamakan '*Konversi Umum*'. Listing program selengkapnya disajikan dalam Lampiran 2.

Sebelum menjalankan program makro '*Konversi Umum*', maka data hasil penelitian yang berasal dari kuesioner dientri ke dalam lembar kerja Minitab terlebih dahulu (Lampiran 3).

Misalkan program di atas disimpan pada direktori f pada folder makro dengan nama konverUm.txt. Untuk mengeksekusinya, perlu menuliskan '%f:\makro\konverum.txt' setelah muncul MTB> pada *session window* Minitab seperti pada Lampiran 4. Hasil dari eksekusi terdiri dari dua buah, yaitu pada lembar sesional dan lembar kerja Minitab seperti terlihat pada Lampiran 5 dan Lampiran 6. Contoh data asli dalam bentuk skala ordinal dan data hasil konversi dalam bentuk interval terlihat pada Lampiran 7 dan Lampiran 8. Data dalam bentuk skala interval tersebut, selanjutnya bisa dimodelkan dengan analisis path (*path analysis*).

Dibandingkan dengan program-program sebelumnya, program '*KonversiUmum*' bisa mengolah data ordinal dengan Skala Likert minimal tiga, misalnya lima, tujuh, sembilan, sebelas, dan seterusnya. Program ini tidak hanya berjalan untuk data ordinal dengan lima kategori terisi penuh seperti program SCL_Ver1, tapi berlaku juga untuk kategori data tidak terisi penuh, asalkan banyaknya kategori yang terisi minimal tiga.

Walau program konversi ini sudah bisa mengatasi kendala program sebelumnya, akan tetapi program ini masih ada kelemahannya yakni perhitungan untuk data asli dengan tiga kategori dibedakan dengan perhitungan untuk data asli dengan lebih dari tiga kategori. Hal ini ditunjukkan oleh program baris kesembilan sampai baris ke-32 Lampiran 3. Perbedaan ini sebagai kelemahan dari program

'KonversiUmum', karena menjadi kurang efisien. Kelemahan lainnya dari program ini adalah tidak bisa melakukan konversi bila jawaban seluruh responden yang diteliti hanya terjawab dua pilihan dari skala Likert yang seharusnya diisi (3 pilihan, 5 pilihan, dan seterusnya).

KESIMPULAN DAN SARAN

Pada penelitian sosial, sering diperlukan transformasi data ordinal ke data interval. Sedangkan perangkat lunak statistika yang ada saat ini belum ada yang menyediakan fasilitas ini secara khusus. Jadi, untuk melakukan transformasi data tersebut, analis/peneliti perlu membuat program dengan bahasa pemrograman tertentu atau program yang bisa dipanggil pada perangkat lunak statistika yang ada, misalnya dengan menggunakan makro Minitab.

Program 'KonversiUmum' sudah berlaku umum tidak terbatas hanya pada data ordinal hasil jawaban responden dengan skala 1-5 dan harus terisi seluruhnya, akan tetapi dapat dijalankan untuk data ordinal mulai dari skala 1-3, 1-5, 1-7 dan seterusnya. Dengan demikian para analis/peneliti lebih leluasa menentukan skala ordinal yang digunakan dalam penelitian dan dituangkan pada jawaban kuesioner penelitian.

Program ini masih perlu disempurnakan lebih lanjut, antara lain karena kurang efisien dalam hal penulisan program dan belum bisa mentransformasi jika pilihan jawaban hanya terisi dua kategori pilihan. Diharapkan ada kajian lebih lanjut untuk membuat program baru yang berlaku umum dan efisien.

DAFTAR PUSTAKA

- Emory, C.W dan Cooper,D.R, 1993. *Business Research Methods*. Irwin. Boston.
- Hays, W. L, 1976. *Quantification in Psychology*. Prentice Hall. New Delhi.
- Muchlis, R. D, 2001. Jurnal. *Penggunaan Makro Minitab untuk Transformasi Data Ordinal ke Data Interval*. Statistika FMIPA UNISBA.
- Nazir, M, 1988. *Metode Penelitian*. Ghalia Indonesia.
- Singarimbun, dan Effendi,S. 1995. *Metode Penelitian Survei*. PT Pustaka LP3ES Indonesia. Jakarta.

Lampiran :

Lampiran 1. Program SCL_Ver1

```
tally ck1;
store c100 c101.
let c102=c101/count(ck1)
copy c102 k3-k7
let k8=k3
let k9=k3+k4
let k10=k3+k4+k5
let k11=k3+k4+k5+k6
let k12=k3+k4+k5+k6+k7
copy k8-k12 c103
cdf c103 c104;
normal 0.0 1.0.
invcdf c103 c104;
normal 0.0 1.0.
pdf c104 c105;
normal 0.0 1.0.
copy c103 k3-k7
copy c105 k8-k12
let k13=(0-k8)/(k3-0)
let k14=(k8-k9)/(k4-k3)
let k15=(k9-k10)/(k5-k4)
let k16=(k10-k11)/(k6-k5)
let k17=(k11-0)/(1-k6)
copy k13-k17 c106
let c107=c106+1+(-1*min(c106))
name c100 'ordinal' c101 'frek' c102 'prop' c103 'prop_kum' c104 'z_val' c105
'z*_val' c106 'sv' c107 'interval'
prin c100-c107
copy c107 k18-k22
code (1) k18 (2) k19 (3) k20 (4) k21 (5) k22 ck1 ck1
let k1=k1+1
end
```

Lampiran 2. Listing Program (KonverUm.TXT)

```
1. gmacro
2. konversi
3. do k1=1:7
4. tally ck1;
5. store c100 c101.
6. let k2=count(ck1)
7. let c102=c101/k2
8. let k3=count(c102)
9. if k3=3
10. copy c102 k3-k5
11. let k8=k3
12. let k9=k3+k4
13. let k10=k3+k4+k5
14. copy k8-k10 c103
15. cdf c103 c104;
16. normal 0.0 1.0.
17. invcdf c103 c104;
18. normal 0.0 1.0.
19. pdf c104 c105;
20. normal 0.0 1.0.
21. copy c103 k3-k5
22. copy c105 k8-k10
23. let k13=(0-k8)/(k3-0)
24. let k14=(k8-k9)/(k4-k3)
25. let k15=(k9-0)/(1-k4)
26. copy k13-k15 c106
27. let c107=c106+1+(-1*min(c106))
28. name c100 'ordinal' c101 'frek' c102 'prop' c103 'prop_kum' c104 'z_val'
    c105 'z*_val' c106 'sv' c107 'interval'
29. prin c100-c107
30. copy c107 k18-k20
31. code (1) k18 (2) k19 (3) k20 ck1 ck1
32. endif
33. if k3>3
34. let c103(1)=c102(1)
35. do k4=2:k3
36. let c103(k4)=c103(k4-1)+c102(k4)
37. cdf c103 c104;
38. normal 0.0 1.0.
39. invcdf c103 c104;
40. normal 0.0 1.0.
41. pdf c104 c105;
42. normal 0.0 1.0.
43. let k5=count(c103)
44. let k6=count(c105)
45. let c106(1)=(0-c105(1))/(c103(1)-0)
46. let k8=k5-1
47. do k7=2:k8
48. let c106(k7)=(c105(k7-1)-c105(k7))/(c103(k7)-c103(k7-1))
49. enddo
50. let c106(k5)=(c105(k6-1)-0)/(1-c103(k5-1))
51. enddo
```

```

52. let c107=c106+1+(-1*min(c106))
53. name c100 'ordinal' c101 'frek' c102 'prop' c103 'prop_kum' c104 'z_val'
    c105 'z*_val' c106 'sv' c107 'interval'
54. prin c100-c107
55. let k9=count(c107)
56. do k50=1:k2
57. do k51=1:k9
58. if ck1(k50)=c100(k51)
59. let ck1(k50)=c107(k51)
60. endif
61. enddo
62. enddo
63. endif
64. erase c100-c107
65. enddo
endmacro

```

Program makro *minitab* di atas mengandung pengertian sebagai berikut :

- Baris 1 : perintah inialisasi untuk global makro.
- Baris 2 : nama global makro.
- Baris 3 : perintah *looping* 'do' untuk nilai k1 dari 1 sampai 7, dalam hal ini k1 merupakan kolom yang terisi data mentah (*raw data*).
- Baris 4-5 : perintah untuk menghitung frekuensi data pada kolom data mentah dan menyimpan hasilnya pada kolom c100 (berisi kategorinya) dan c101 (berisi frekuensi tiap kategori).
- Baris 6 : memberikan nilai banyaknya baris yang ada pada kolom ck1 pada k2.
- Baris 7 : perintah untuk menghitung proporsi dan disimpan pada kolom c102.
- Baris 8 : memberikan nilai banyaknya baris yang ada pada kolom c102 pada k3.
- Baris 9 : perintah 'if' untuk nilai k3, banyaknya baris pada kolom c102, adalah 3.
- Baris 10 : perintah mengcopy nilai proporsi yang ada pada kolom c102 ke dalam k3 sampai k5.
- Baris 11-14 : perintah menghitung proporsi kumulatif dan disimpan pada kolom c103.
- Baris 15-16, 37-38 : perintah untuk menghitung fungsi kepekatan peluang dari sebaran normal baku dan hasilnya disimpan pada kolom c104.
- Baris 17-18, 39-40 : perintah untuk menghitung invers dari fungsi sebaran kumulatif normal baku dari kolom c103 dan ditampilkan pada kolom c104.
- Baris 19-20, 41-42 : perintah untuk menghitung nilai fungsi peluang dari kolom c104 dan ditampilkan pada kolom c105.
- Baris 21 : perintah menyimpan nilai yang ada pada kolom c103 pada konstanta k3 sampai k5.
- Baris 22 : perintah menyimpan nilai yang ada pada kolom c105 pada konstanta k8 sampai k10.
- Baris 23-25 : perintah untuk menghitung *scale value* untuk kategori satu sampai tiga, dan hasilnya disimpan pada k13 sampai k15.
- Baris 26 : perintah untuk memindahkan konstanta k13 sampai k15 ke kolom c106.
- Baris 27,52 : menghitung *score* untuk setiap item pada kolom c106 dan hasilnya disimpan pada kolom c107.

- Baris 28,53 : perintah untuk memberi nama kolom c100 sampai kolom c107. Kolom c100 dinamai 'ordinal', kolom c101 dinamai 'frek', kolom c102 dinamai 'prop', kolom c103 dinamai 'prop_kum', kolom c104 dinamai 'z_val', kolom c105 'z*_val', c106 'sv', dan kolom c107 'interval'.
- Baris 29,54 : perintah untuk mencetak kolom c100 sampai c107.
- Baris 30-31 : perintah untuk mengganti data ordinal pada kolom ck1 dengan hasil score yang diperoleh, hasilnya disimpan pada kolom ck1.
- Baris 32, 63 : perintah untuk mengakhiri perintah 'if'.
- Baris 33 : perintah 'if' untuk banyaknya baris pada kolom c102 lebih dari 3.
- Baris 34 : proporsi kumulatif untuk kategori pertama.
- Baris 35-36 : perintah menghitung proporsi kumulatif dan disimpan pada kolom c103.
- Baris 43 : menyimpan banyaknya baris pada kolom c103 pada k5.
- Baris 44 : menyimpan banyaknya baris pada kolom c105 pada k6.
- Baris 45 : menghitung scale value untuk kategori pertama dan disimpan pada kolom c106 baris pertama.
- Baris 46 : menyimpan nilai banyaknya baris pada kolom c103 dikurangi satu ke k8.
- Baris 47-49 : menghitung scale value untuk kategori kedua sampai kategori sebelum yang terakhir dan disimpan pada kolom c106.
- Baris 50 : menghitung scale value untuk kategori terakhir dan disimpan pada kolom c106 dan baris terakhir.
- Baris 51, 61, 62, 65 : perintah untuk mengakhiri perintah 'do'.
- Baris 55-62 : perintah untuk mengganti data ordinal pada kolom ck1 dengan score dari kolom c107.
- Baris 64 : perintah untuk menghapus kolom c100 sampai c107.
- Baris 65 : perintah untuk mengakhiri global makro.

Lampiran 3. Tampilan input data pada lembar kerja Minitab

	C1	C2	C3	C4	C5	C6	C7
↓	X111	X112	X121	X122A	X122B	X123	X131
1	4	4	5	4	4	1	4
2	4	4	4	4	4	1	4
3	4	4	3	4	5	5	4
4	5	5	4	5	4	4	5
5	4	4	5	4	4	4	5
6	5	5	3	5	4	1	2
7	5	5	4	5	5	1	5
8	5	5	5	5	4	1	5
9	5	5	1	5	5	4	4
10	4	4	4	4	4	4	5
11	3	3	4	4	4	4	5
12	4	4	4	5	4	4	5
13	4	4	4	4	3	4	4
14	3	4	4	4	5	1	2
15	1	5	1	5	5	1	4

Lampiran 4. Tampilan eksekusi makro pada lembar sesional

```

MINITAB - Untitled Worksheet - [Session]
File Edit Manip Calc Stat Graph Editor Window Help

Worksheet size: 100000 cells

MTB > %f:\pakirjen\makro\konverum.txt
    
```

Lampiran 5. Tampilan output pada lembar sesional.

Worksheet size: 100000 cells

MTB > %f:\pakirjen\makro\konverum.txt
Executing from file: f:\pakirjen\makro\konverum.txt.

Summary Statistics for Discrete Variables

X111	Count
1	1
3	3
4	27
5	10
N=	41

* NOTE * Arithmetic fault, MISSING returned 1 times.
* Value out of bounds.

Data Display

Row	ordinal	frek	prop	prop_kum	z_val	z*_val	sv
1	1	1	0.024390	0.02439	-1.97051	0.057247	-2.34712
2	3	3	0.073171	0.09756	-1.29557	0.172356	-1.57315
3	4	27	0.658537	0.75610	0.69380	0.313605	-0.21449
4	5	10	0.243902	1.00000	*	*	1.28578

interval
1.00000 1.77396 3.13263 4.63290

Summary Statistics for Discrete Variables

Lampiran 6. Tampilan output pada lembar kerja Minitab.

	C1	C2	C3	C4	C5	C6	C7
↓	X111	X112	X121	X122A	X122B	X123	X131
1	3.13263	3.12543	4.18422	4.00000	3.14312	1.00000	3.41223
2	3.13263	3.12543	2.96355	4.00000	3.14312	1.00000	3.41223
3	3.13263	3.12543	1.96515	4.00000	4.52169	3.69933	3.41223
4	4.63290	4.69427	2.96355	5.00000	3.14312	2.41847	4.63290
5	3.13263	3.12543	4.18422	4.00000	3.14312	2.41847	4.63290
6	4.63290	4.69427	1.96515	5.00000	3.14312	1.00000	2.12496
7	4.63290	4.69427	2.96355	5.00000	4.52169	1.00000	4.63290
8	4.63290	4.69427	4.18422	5.00000	3.14312	1.00000	4.63290
9	4.63290	4.69427	1.00000	5.00000	4.52169	2.41847	3.41223
10	3.13263	3.12543	2.96355	4.00000	3.14312	2.41847	4.63290
11	1.77396	1.67302	2.96355	4.00000	3.14312	2.41847	4.63290
12	3.13263	3.12543	2.96355	5.00000	3.14312	2.41847	4.63290
13	3.13263	3.12543	2.96355	4.00000	2.03291	2.41847	3.41223
14	1.77396	3.12543	2.96355	4.00000	4.52169	1.00000	2.12496
15	1.00000	4.69427	1.00000	5.00000	4.52169	1.00000	3.41223

Lampiran 7. Contoh *raw data* dengan skala ordinal

X111	X112	X121	X122A	X122B	X123	X131
4	4	5	4	4	1	4
4	4	4	4	4	1	4
4	4	3	4	5	5	4
5	5	4	5	4	4	5
4	4	5	4	4	4	5
5	5	3	5	4	1	2
5	5	4	5	5	1	5
5	5	5	5	4	1	5
5	5	1	5	5	4	4
4	4	4	4	4	4	5
3	3	4	4	4	4	5
4	4	4	5	4	4	5

Lampiran 8. Contoh data hasil transformasi.

X111	X112	X121	X122A	X122B	X123	X131
3.13263	3.12543	4.18422	4.00000	3.14312	1.00000	3.41223

3.13263	3.12543	2.96355	4.00000	3.14312	1.00000	3.41223
3.13263	3.12543	1.96515	4.00000	4.52169	3.69933	3.41223
4.63290	4.69427	2.96355	5.00000	3.14312	2.41847	4.63290
3.13263	3.12543	4.18422	4.00000	3.14312	2.41847	4.63290
4.63290	4.69427	1.96515	5.00000	3.14312	1.00000	2.12496
4.63290	4.69427	2.96355	5.00000	4.52169	1.00000	4.63290
4.63290	4.69427	4.18422	5.00000	3.14312	1.00000	4.63290
4.63290	4.69427	1.00000	5.00000	4.52169	2.41847	3.41223